

CAM superelliptische beker voor extrusie : listing - technical manual

Citation for published version (APA):

Muns, J. C. (1989). *CAM superelliptische beker voor extrusie : listing - technical manual*. (TH Eindhoven. Afd. Werktuigbouwkunde, Vakgroep Produktietechnologie : WPB; Vol. WPA0681). Technische Universiteit Eindhoven.

Document status and date:

Gepubliceerd: 01/01/1989

Document Version:

Uitgevers PDF, ook bekend als Version of Record

Please check the document version of this publication:

- A submitted manuscript is the version of the article upon submission and before peer-review. There can be important differences between the submitted version and the official published version of record. People interested in the research are advised to contact the author for the final version of the publication, or visit the DOI to the publisher's website.
- The final author version and the galley proof are versions of the publication after peer review.
- The final published version features the final layout of the paper including the volume, issue and page numbers.

[Link to publication](#)

General rights

Copyright and moral rights for the publications made accessible in the public portal are retained by the authors and/or other copyright owners and it is a condition of accessing publications that users recognise and abide by the legal requirements associated with these rights.

- Users may download and print one copy of any publication from the public portal for the purpose of private study or research.
- You may not further distribute the material or use it for any profit-making activity or commercial gain
- You may freely distribute the URL identifying the publication in the public portal.

If the publication is distributed under the terms of Article 25fa of the Dutch Copyright Act, indicated by the "Taverne" license above, please follow below link for the End User Agreement:

www.tue.nl/taverne

Take down policy

If you believe that this document breaches copyright please contact us at:

openaccess@tue.nl

providing details and we will investigate your claim.

BIB 400000

CAM Superelliptische beker voor Extrusie

Listing - Technical manual

J.C. Muns

januari 1989
VF-code D2

HSV-Venlo stageverslag
WPA 0681

~~VEIBOUWELIJK~~

Periode : 1-12-'88 - 28-02-'89

Bedrijfsbegeleider: Ir. L.J.A. Houtackers
Schoolbegeleider : Ir. P.B.G. Peeters

Plaats: Laboratorium voor Omvormtechnologie
Vakgroep WPA- TUE Eindhoven.

In opdracht van HSV HTS Venlo.

Inhoudsopgave

	blz.
CNCBeker.pas	1
Function power	4
Function nroot	5
Function ellips	6
Procedure coördinaten	7
Procedure nauwkeurigheid	8
Hoofdprogramma	9
Berekenen van de coördinaten voor de matrijs	10
Berekenen van de coördinaten voor de stempel	11
Opmerking	12

CNCBeker.pas

CNCBeker.pas is een programma dat de berekende coördinaten voor de superelliptische stempel en matrijs naar de volgende files wegschrijft:

```
b:\mna      (coördinaten voor de matrijs)
b:\sna      (coördinaten voor de stempel)
```

Elke regel (1 x-coördinaat met bijbehorende y-coördinaat) wordt voorzien van een regelnummer en de CNC bewerking G1 (lineaire interpolatie).

Bij het gebruik van dit programma moet dus een andere diskette in drive b gestoken worden. Wil men de coördinaten naar een andere naam of drive wegschrijven dan moet in deze source file de 'pathname' in de betreffende regel veranderd worden.

Beschikt men alleen over de execute versie van dit programma dan dient men gebruik te maken van het ms-dos commando assign. Dit assign commando wordt dan op de juiste manier in de batch file voor het opstarten van cncbeker.pas geplaatst.

```

(*****
*
*   titel      : GEOMETRIE van SUPERELLIPTISCHE BEKER voor EXTRUSIE.
*
*
*   module    :
*               (Maakt gebruik van: grafunit, HK3 en cursor
*               28 augustus 1988
*               Adri Melis.)
*
*
*   versie    : 2.00
*   datum     : 15-1-'89
*
*
*   language  : Turbo Pascal version 4.0 (Borland)
*
*
*   systeem   : IBM - XT met EGA-kaart
*
*
*   programmeur: Jaap Muns
*
*
*   informatie : Technische Universiteit Eindhoven
*               Faculteit der Werktuigbouwkunde
*               Vakgroep Produktietechnologie en -Automatisering
*               Laboratorium voor Omvormtechnologie
*
*
*   telefoon  : 040 - 474521 (ir. L.J.A. Houtackers)
*
*
*               077 - 731949 (J.C. Muns)
*****)

```

```

program bekerCNC;

```

```

uses HK3, printer;           {van deze units maakt dit programma gebruik}

```

```

const st = 0.1;             {stapgrootte waarmee de y-waarde wordt opgehoogd}
      C = 1000;             {constante voor de array grootte}

```

```

type TRij = Array [0..C] of real; {array waarin de coördinaten opgeslagen worden}
TBewerken = (na);               {na staat voor nadraaien, er moet naar de nauwkeurig-
                                heidsprocedure gesprongen worden}
TContour = (binnen, buiten);   {geeft aan of de binnen- of buitencontour
                                berekend moet worden}
woord = string[255];           {string van maximaal 255 tekens}

```

```

Tcoördinaten=text;

```

```

var dmax : real;              {diameter van matrijs (buitendiameter van beker)}
    hd   : real;              {matrijshoogte}
    h    : real;              {totale hoogte}
    b2   : real;              {diepte van de beker}
    n1, n2: integer;          {exponenten van de buiten resp. binnencontour}
    a1   : real;              {a1= straal van de matrijs (dmax/2)}
    a2   : real;              {a2= maximale straal van het stempel}
    sb   : real;              {bodemdikte van de beker}
    sw   : real;              {wanddikte aan de bovenzijde van de beker}
    x1, x2,
    z1, z2: TRij;            {y coördinaten, z genoemd vanwege CNC assenstelsel}
    aant1, aant2: integer;    {aantal intervallen}

```

```
i, j : integer;           {tellers}
key   : string[1];       {er wordt gewacht tot u op een toets drukt}
xn, zn: TRij;            {array met nieuwe x en z-coördinaten}
xwaarde, zwaarde: string[255]; {conversie van een real naar een string variabele}
k, kmax : integer;       {tellers}
bewerken: TBewerken;
fil      : Tcoördinaten; {hele CNC regel met regelnummer, bewerking G1, X- en
                          Z-coördinaten wordt steeds opgeslagen in een tekstfile}
line : woord;            {tussentijdse waarden van de CNC regel die op dat
                          moment bekend zijn}
x, z : woord;            {de letters X en Z in de CNC regel}
p     : integer;         {teller voor het regelnummer voor het CNC programma}
regelnummer: string[5]; {conversie van een integer naar een string variabele}
```

Function power

Omdat pascal geen standaardfunctie kent voor het berekenen van machten is deze function geschreven.

Er wordt gebruik gemaakt van de volgende wiskundige definitie:

$$x^n \quad \langle == \rangle$$

$$e^{\ln(x^n)} \quad \langle == \rangle$$

$$e^{n \ln(x)}$$

x : constante of vergelijking
n : exponent
e : standaardfunctie in pascal
ln: standaardfunctie in pascal

De exponent n moet bij de function power een integer zijn.

```
function power (x: real; n: integer): real;  
{prec: waarden x en n moeten bekend zijn.  
postc: power heeft de waarde x^n .      }
```

```
begin  
  if (x <> 0)  
    then begin  
      if ((n mod 2) = 0)  
        then power:= exp(n * ln (abs(x)))  
        else  
          if (x > 0) then power:= exp(n * ln(abs(x)))  
          else power:= -1 * exp(n * ln(abs(x)))  
        end {then}  
      else if n = 0 then power:= 1  
        else power:= 0;  
    end; {of function power}
```

Function root

Voor de function root geldt hetzelfde als de function power.
Nu kan echter ook gerekend worden met real machten.

```
function root (x: real; n: integer): real;
{prec: waarden x en n moeten bekend zijn.
 postc: root heeft de waarde  $x^{1/n}$  . }

var m: real;

begin
  m:=1/n;
  if x <> 0
  then begin
    if ((n mod 2) = 0)
    then nroot:= exp(m * ln(abs(x)))
    else
      if (x>0) then nroot:= exp(m * ln (abs(x)))
      else nroot:= -1 * exp(m * ln (abs(x)))
    end {then}
  else if n=0
  then write ('x^(1/n) is oneindig groot voor n= 0.')
  else nroot:= 1;
end; {of function root}
```


Function ellips

In deze function zijn de functions power en nroot zodanig verwerkt dat de coördinaten van de superellips berekend worden.

```
function ellips (y, a, b: real; n: integer): real;
{ prec: y, a, b en n zijn bekend.
  postc: functiewaarde x van opgegeven superellips is berekend.}

var hulp: real;

begin
  hulp:= 1 - power((1 - y/b), n);
  ellips:= a * nroot(hulp, n);      {x-waarde van superellips is berekend}
end; {of function ellips}
```

Procedure coördinaten

Het deel tot hoogte h_D wordt opgedeeld in intervallen (y-waarden) ter grootte van 0.1 mm (st). Bij elke y-waarde wordt door het aanroepen van function ellips de bijbehorende x-coördinaat berekend.

De berekende coördinaten (van het superelliptische deel) van de buiten- en binnencontour worden in een array geplaatst.

```
procedure coördinaten (var x, y: TR1j; var nint: integer;
                      a, hd, stap,s: real; n: integer; contour:TContour);
```

```
{ prec: a, hd, stap(=stapgrootte), s (bodemdikte), n en contour zijn bekend.
  postc: 2 array's zijn gevuld met x en y. }
```

```
var hulp:real;
    i, j: integer;
```

```
begin
  nint:= round(hd/stap);
  stap:= hd/nint;
  i:= 0;
  y[i]:= 0;
  x[i]:= 0;
  while (x[i] < (a-0.000000000001)) and (i <= nint) do
  begin
    i:= i + 1;
    y[i]:= y[i-1] + stap;
    x[i]:= ellips (y[i], a, hd, n);
  end; {while}
  if (i <= nint) then
  begin
    for j:= i to nint do
    begin
      x[j]:= a;
      y[j]:= y[j-1] + stap;
    end; {next j}
  end; {then}
  if contour=binnen then for i:=0 to nint do y[i]:= y[i] + s;
end; {of procedure coördinaten}

{voor de binnencontour wordt bij
elke y-coördinaat uit de array
de bodemdikte opgeteld. Zo wordt
op een eenvoudige manier
de y-coördinaat met bijbehorende
x-coördinaat voor de binnen-
contour berekend.}
```

Procedure Nauwkeurigheid

Deze procedure is geschreven omdat de x-waarde in de kromming van de superellips sterk verandert bij een kleine verandering van de y-waarde. Dit verschijnsel doet zich vooral voor bij een hogere exponent n (rechthoekiger vorm).

Wanneer twee opeenvolgende x-coördinaten minder dan 0.001 verschillen wordt de laatste x-coördinaat van deze twee overgeslagen. Er wordt dus gesprongen naar de volgende x-coördinaat.

Dit gaat net zo lang door totdat het verschil groter is dan 0.001, de coördinaat waar dat voor geldt wordt weer naar de file weggeschreven. Op deze manier wordt dus voorkomen dat alle berekende coördinaten naar een file weggeschreven worden. In de file zouden anders een aantal overbodige regels staan (verschil x-coördinaten < 0.001), die toch niet zo precies door de CNC draaibank gemaakt kunnen worden.

De CNC draaibank werkt in drie decimalen, maar of zo'n nauwkeurigheid te realiseren valt hangt ook af van:

de beitelpunt
of werkstuk en beitel lichtelijk trillen of slingeren

```
procedure Nauwkeurigheid (var xn,zn: TRij; var k:integer;
    x, z:TRij; nint:integer; bewerken:TBewerken; contour:TContour);

{prec: x, z, bewerken en contour zijn bekend.
 postc: twee nieuwe arrays xn en zn zijn gevuld met x en z coördinaten
        zodanig dat bij twee opeenvolgende z-coördinaten de bijbehorende
        x-coördinaten op zijn minst 0.001 mm van elkaar verschillen.    }

var i      : integer;
    hulp   : real;
    stapx  : real;

begin
    if (bewerken=na) then stapx:= 0.001;
    k:= 0;
    i:= 1;
    hulp:=x[0];
    xn[k]:= x[0];
    if (contour=buiten) then zn[k]:= z[0]
        else zn[k]:= -z[0];
    repeat
        while (((x[i]-hulp) < stapx) and (i<nint)) do i:= i+1;    {verschil x-coördinaten < 0.001,
                                                                    sprong naar volgende coördinaat}
        k:= k+1;
        xn[k]:= 2*x[i];          {omzetten van radius naar diameter}
        if (contour=buiten) then zn[k]:= z[i]
            else zn[k]:= -z[i];
        hulp:= x[i];
        i:= i+1;
    until (i>nint);
end; {of procedure Nauwkeurigheid}
```

```

(hoofdprogramma)

begin
write ('Voer de uitwendige diameter in: (10',+chr(243),'dmax',+chr(243),'100)');
write ('          dmax: '); readln (dmax);
while ((dmax<10) or (dmax>100)) do
begin
write ('Voer de uitwendige diameter opnieuw in: (10',+chr(243),'dmax',+chr(243),'100)');
write ('          dmax: '); readln (dmax);
end; {while}
write ('Voer de totale hoogte in: (5',+chr(243),'h',+chr(243),'200)');
write ('          h   : '); readln (h);
while (h<5) or (h>200) do
begin
write ('Voer de totale hoogte opnieuw in: (5',+chr(243),'h',+chr(243),'100)');
write ('          h   : '); readln (h);
end; {while}
write ('Voer de matrijs-hoogte in: (0<hd',+chr(243),'0.9*h)');
write ('          hd  : '); readln (hd);
while (hd<=0) or (hd>0.9*h) do
begin
write ('Voer de matrijs-hoogte opnieuw in: (0<hd',+chr(243),'0.9*h)');
write ('          hd  : '); readln (hd);
end; {while}
write ('Voer de wanddikte in: (0<sw',+chr(243),'0.1*dmax, en sw<hd)');
write ('          sw  : '); readln (sw);
while (sw<=0) or (sw>0.1*dmax) or (sw>=hd) do
begin
write ('Voer de wanddikte opnieuw in: (0<sw',+chr(243),'0.1*dmax, en sw<hd)');
write ('          sw  : '); readln (sw);
end; {while}
write ('Voer de bodemdikte in: (sb>sw en sb<hd)');
write ('          sb  : '); readln (sb);
while (sb<=sw) or (sb>=hd) do
begin
write ('Voer de bodemdikte opnieuw in: (sb>sw en sb<hd)');
write ('          sb  : '); readln (sb);
end; {while}
writeln ('Voer de exponent voor de buitencontour in: ');
write ('          (geheel getal tussen 2 en 40)          : '); readln (n1);
while (n1<2) or (n1>40) do
begin
writeln ('Voer de exponent voor de buitencontour opnieuw in: ');
write ('          (geheel getal tussen 2 en 40)          : '); readln (n1);
end; {while}
writeln ('Voer de exponent voor de binnencontour in: (n2',+chr(243),'n1)');
write ('          (geheel getal tussen 2 en n1)          : '); readln (n2);
while (n2<2) or (n2>n1) do
begin
writeln ('Voer de exponent voor de binnencontour opnieuw in: (n2',+chr(243),'n1)');
write ('          (geheel getal tussen 2 en n1)          : '); readln (n2);
end; {while}
writeln;
writeln;

```

```

{berekenen van de coördinaten voor de matrijs}

a1:= dmax/2;
a2:= a1-sw;
x:= 'X';           {de letter X}
z:= 'Z';           {de letter Z}

coördinaten (x1,z1,aant1,a1,hd,st,0,n1,buiten);
bewerken:= na;
Nauwkeurigheid (xn,zn,kmax,x1,z1,aant1,bewerken,buiten);
assign(fil,'b:\Mna');           {de file Mna wordt op drive b aangemaakt}
rewrite(fil);
p:=480;                          {eerste regelnummer CNC regel}
for k:=kmax downto 0 do          {de berekende coördinaten worden in omgekeerde
                                volgorde afgedrukt i.v.m. het nulpunt van het
                                CNC programma}

begin
  Str(p,regelnummer);
  line:='N' + regelnummer + ' G1 '; {CNC regel met regelnummer en bewerking G1}
  write(fil, line);
  write (fil, x);                  {de letter X wordt in de regel geplaatst}
  Str(xn[k]:4:3, xwaarde);
  line := xwaarde;
  write (fil, line);               {de x-coördinaat wordt in de regel geplaatst}
  write (fil, z);                 {de letter Z wordt in de regel geplaatst}
  Str((zn[k]-hd-5):4:3, zwaarde);  {de berekende z-waarde wordt verminderd
  met hd i.v.m. het nulpunt
  van het CNC-programma}

  line:= zwaarde;
  writeln (fil, line);            {de z-coördinaat wordt in de regel geplaatst
  de eerste CNC regel is nu compleet}

  p:=p+1;
end; {next k}
writeln('*****');
writeln ('* De file Mna bevat de coördinaten voor het nabewerken van de matrijs *');
writeln('*****');
writeln;
reset (fil);
while not eof(fil) do begin      {de file met coördinaten wordt ook zichtbaar
  readln (fil, line); writeln (line);  gemaakt op het scherm}
end; {while}
close(fil);
writeln ('geef return voor vervolg'); readln (key);

```

```

{berekenen van de coördinaten voor de stempel}

b2:= hd-sb;
coördinaten (x2,z2,aant2,a2,b2,st,0,n2,binnen);
bewerken:= na;
Nauwkeurigheid (xn,zn,kmax,x2,z2,aant2,bewerken,binnen);
assign(fil,'b:\Sna');          (de file Sna wordt op drive b aangemaakt)
rewrite(fil);
p:=360;
for k:=0 to kmax do
  begin
    Str(p,regelnummer); line:= 'N' + regelnummer + ' G1 ';
    write(fil, line);
    write (fil, x);
    Str(xn[k]:4:3, xwaarde); line := xwaarde;
    write (fil, line);
    write (fil, z);
    Str(zn[k]:4:3, zwaarde); line:= zwaarde;
    writeln (fil, line);
    p:=p+1;
  end; {next k}
writeln('*****');
writeln ('* De file Sna bevat de coördinaten voor het nabewerken van het stempel *');
writeln('*****');
writeln;
reset (fil);
while not eof(fil) do begin
  readln (fil, line); writeln (line);
end; {while}
close(fil);
end.

```

Opmerking

Dit programma maakt gebruik van dezelfde invoerwaarden als het berekeningsprogramma Bemoment.pas. Ook wordt gebruik gemaakt van een aantal dezelfde functions en procedures. Daarom kan dit programma ook in het berekeningsprogramma ingepast worden.